<commit_before>{ config, lib, pkgs, options }:

with lib;

let
  cfg = config.services.prometheus.exporters.smartctl;
  format = pkgs.formats.yaml {};
  configFile = format.generate "smartctl-exporter.yml" {
    smartctl_exporter = {
      bind_to = "${cfg.listenAddress}:${toString cfg.port}";
      url_path = "/metrics";
      smartctl_location = "${pkgs.smartmontools}/bin/smartctl";
      collect_not_more_than_period = cfg.maxInterval;
      devices = cfg.devices;
    };
  };
in {
  port = 9633;

  extraOpts = {
    devices = mkOption {
      type = types.listOf types.str;
      default = [];
      example = literalExpression ''
        [ "/dev/sda", "/dev/nvme0n1" ];
      '';
      description = ''
        Paths to the disks that will be monitored. Will autodiscover
        all disks if none given.
      '';
    };
    maxInterval = mkOption {
      type = types.str;
      default = "60s";
      example = "2m";
      description = ''
        Interval that limits how often a disk can be queried.
      '';
    };
  };

  serviceOpts = {
    serviceConfig = {
      AmbientCapabilities = [
        "CAP_RAW_SYSIO"
        "CAP_SYS_ADMIN"
      ];
      CapabilityBoundingSet = [
        "CAP_RAW_SYSIO"
        "CAP_SYS_ADMIN"
      ];
      DevicePolicy = "closed";
      DeviceAllow = lib.mkOverride 100 (
        if cfg.devices != [] then
          cfg.devices
        else [
          "block-blkext rw"
          "block-sd rw"
          "char-nvme rw"
        ]
      );
      ExecStart = ''
        ${pkgs.prometheus-smartctl-exporter}/bin/smartctl_exporter -config ${configFile}
      '';
      PrivateDevices = lib.mkForce false;
      ProtectProc = "invisible";
      ProcSubset = "pid";
      SupplementaryGroups = [ "disk" ];
      SystemCallFilter = [
        "@system-service"
        "~@privileged @resources"
      ];
    };
  };
}
<commit_msg>nixos/smartctl-exporter: fix typo in rawio capab<commit_after>{ config, lib, pkgs, options }:

with lib;

let
  cfg = config.services.prometheus.exporters.smartctl;
  format = pkgs.formats.yaml {};
  configFile = format.generate "smartctl-exporter.yml" {
    smartctl_exporter = {
      bind_to = "${cfg.listenAddress}:${toString cfg.port}";
      url_path = "/metrics";
      smartctl_location = "${pkgs.smartmontools}/bin/smartctl";
      collect_not_more_than_period = cfg.maxInterval;
      devices = cfg.devices;
    };
  };
in {
  port = 9633;

  extraOpts = {
    devices = mkOption {
      type = types.listOf types.str;
      default = [];
      example = literalExpression ''
        [ "/dev/sda", "/dev/nvme0n1" ];
      '';
      description = ''
        Paths to the disks that will be monitored. Will autodiscover
        all disks if none given.
      '';
    };
    maxInterval = mkOption {
      type = types.str;
      default = "60s";
      example = "2m";
      description = ''
        Interval that limits how often a disk can be queried.
      '';
    };
  };

  serviceOpts = {
    serviceConfig = {
      AmbientCapabilities = [
        "CAP_SYS_RAWIO"
        "CAP_SYS_ADMIN"
      ];
      CapabilityBoundingSet = [
        "CAP_SYS_RAWIO"
        "CAP_SYS_ADMIN"
      ];
      DevicePolicy = "closed";
      DeviceAllow = lib.mkOverride 100 (
        if cfg.devices != [] then
          cfg.devices
        else [
          "block-blkext rw"
          "block-sd rw"
          "char-nvme rw"
        ]
      );
      ExecStart = ''
        ${pkgs.prometheus-smartctl-exporter}/bin/smartctl_exporter -config ${configFile}
      '';
      PrivateDevices = lib.mkForce false;
      ProtectProc = "invisible";
      ProcSubset = "pid";
      SupplementaryGroups = [ "disk" ];
      SystemCallFilter = [
        "@system-service"
        "~@privileged @resources"
      ];
    };
  };
}

{'fraction_non_alphanumeric': 0.1249669749009247, 'fraction_numerical': 0.006340819022457068, 'mean_word_length': 2.0047619047619047, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1918748', 'n_tokens_mistral': 1250, 'n_tokens_neox': 1208, 'n_words': 239}